
===========RMS_ANY_SIGNAL=======================

Because MacSpice stores results as arrays, it is not hard
to do the math processing to find the RMS of a signal. 

RMS_Input_Signals
* dsauersanjose@aol.com  2.10.10_1.36PM
* www.idea2ic.com 
*                         the spectrum output appears to have..
*                         1Vpk cosine => real_mag 1 => real_rms .707
*     Vtime       ^  OUT  1Vpk sine   => imag_mag 1 => imag_rms .707
*        ____    /_\ 
*  VT  _|_   |   _|_      frequency(0) = out of range 
*     /_  \  |  /_  \     (No DC term )
*    // \  \ |_// \  \ 
*    \   \//   \   \// 
*     \___/     \___/ B1 
*       |         | 
*      _|_       _|_ 
*      ///       /// 
* 
.Option   srcsteps = 1     set Gmin = 1.0000E-02
*=========Circuit_Netlist=========================
VT        Vtime      0     PWL    ( 0 0 1 1 ) DC 0
B1C       OUT1C      0     V =    cos(2*3.14159*5*v(Vtime))  
B1S       OUT1S      0     V =    1.4142*sin(2*3.14159*5*v(Vtime))
*================================================================

Creating waveforms in the above format can make it 
easier to see critical details like amplitude, 
frequency, and phase.

*=========Clipping===============================================
B1CC      VOUT1C     0     V =    2*u( v(OUT1C))-1
*================================================================

The u function is basically a sign function.

*=========Transient==============================================
*TRAN     TSTEP      TSTOP TSTART TMAX   ?UIC?
.tran     0.1m       1     0      0.1m    
.control 
run 
set       pensize =  2 
plot      v(vout1c)   v(out1s)  
*================================================================

Running and plotting is straight forward.

*===========Find_RMS_OUT1S=========================
let num =   length(out1s)-1
let i  =    0
let vpwr =  0
repeat      $&num
let i =     i +1
let vpwr =  vpwr  + (mag(OUT1S[i])*mag(OUT1S[i]))/num
end
let vrms1 = sqrt(vpwr)



echo        INPUT RMS SINE =  $&vrms1
*================================================================

MacSpice stores the plots in an array format.
The size of the array can be found.
And doing a RMS on waveforms is not hard.

*===========Find_RMS_VOUT1C=========================
let num =   length(vout1c)-1
let i  =    0
let vpwr =  0
repeat      $&num
let i =     i +1
let vpwr =  vpwr  + (mag(vout1c[i])*mag(vout1c[i]))/num
end
let vrms1 = sqrt(vpwr)
echo        INPUT RMS SQUARE =  $&vrms1
*================================================================

MacSpice will print out the following.

Circuit: RMS_Input_Signal

INPUT RMS SINE = 0.999496
INPUT RMS SQUARE = 1

*===========Wrap_Up=========================
.endc 
.end 

===========Full_Netlist_For_Copy_Paste=======================
RMS_Input_Signal
* dsauersanjose@aol.com  1/21/10
* www.idea2ic.com 
*                         the spectrum output appears to have..
*                         1Vpk cosine => real_mag 1 => real_rms .707
*     Vtime       ^  OUT  1Vpk sine   => imag_mag 1 => imag_rms .707
*        ____    /_\ 
*  VT  _|_   |   _|_      frequency(0) = out of range 
*     /_  \  |  /_  \     (No DC term )
*    // \  \ |_// \  \ 
*    \   \//   \   \// 
*     \___/     \___/ B1 
*       |         | 
*      _|_       _|_ 
*      ///       /// 
* 
.Option   srcsteps = 1     set Gmin = 1.0000E-02
*=========Circuit_Netlist=========================
VT        Vtime      0     PWL    ( 0 0 1 1 )
B1C       OUT1C      0     V =    cos(2*3.14159*5*v(Vtime))  
B1S       OUT1S      0     V =    1.4142*sin(2*3.14159*5*v(Vtime))
*================================================================
B1CC      VOUT1C     0     V =    2*u( v(OUT1C))-1
*================================================================
*TRAN     TSTEP      TSTOP TSTART TMAX   ?UIC?
.tran     0.1m       1     0      0.1m    
.control 
run 
set       pensize =  2 
plot      v(vout1c)   v(out1s)  
*===========Find_RMS_OUT1S=========================
let num =   length(out1s)-1
let i  =    0
let vpwr =  0
repeat      $&num
let i =     i +1
let vpwr =  vpwr  + (mag(OUT1S[i])*mag(OUT1S[i]))/num
end
let vrms1 = sqrt(vpwr)
echo        INPUT RMS SINE =  $&vrms1
*===========Find_RMS_VOUT1C=========================
let num =   length(vout1c)-1
let i  =    0
let vpwr =  0
repeat      $&num
let i =     i +1
let vpwr =  vpwr  + (mag(vout1c[i])*mag(vout1c[i]))/num
end



let vrms1 = sqrt(vpwr)
echo        INPUT RMS SQUARE =  $&vrms1
.endc 
.end 

